
 Gravação e Alteração de Macros 

 

1. Grave uma macro que execute as seguintes operações. (Tools � Macros � Record New Macro) 
Não se esqueça de terminar a gravação (Tools � Macros � Stop Recording) 
a) escreva o seu nome na célula Activa 
b) à esquerda do seu nome escreva o seu número e o seu curso. 
c) Altere a cor de fundo dessas 3 células para amarelo 
d) Coloque o seu nome a bold. 
e) Ajuste a largura da coluna da primeira célula para que o seu nome caiba na célula 

2. Confirme que terminou a gravação da sua macro. 
Verifique o código produzido, que deve ser mais ou menos igual ao código seguinte. 

 
Sub MinhaInfor() 
' MinhaInfor Macro 
' Macro recorded 08-05-2002 by fmmb 
' 
    ActiveCell.FormulaR1C1 = "António da Silva" 
    ActiveCell.Offset(0, 1).Range("A1").Select 
    ActiveCell.FormulaR1C1 = "20518" 
    ActiveCell.Offset(0, 1).Range("A1").Select 
    ActiveCell.FormulaR1C1 = "OGE" 
    ActiveCell.Offset(0, -2).Range("A1:C1").Select 
    With Selection.Interior 
        .ColorIndex = 6 
        .Pattern = xlSolid 
    End With 
    ActiveCell.Select 
    Selection.Font.Bold = True 
    ActiveCell.Columns("A:A").EntireColumn.EntireColumn.AutoFit 
End Sub 

3. Escreva uma macro que escreva o seu nome na célula activa. Escreva o seu número e Curso nas células 
imediatamente abaixo. 
Sugestão: Copie o código da macro anterior e altere o que for necessário. 

 Execução sequencial. MSGBOX 

Sub exec_1() 
    MsgBox "Viva eu." 
    MsgBox 99 
    MsgBox "Número 33." 
End Sub 
 
Sub exec_3() 
    MsgBox "Hoje é " & Date 
    MsgBox "São justamente " & Time 
    MsgBox "Amanhã será " & Date + 1 
End Sub 

1. Explique o efeito da subrotina exec_1. 
A ordem das instruções é indiferente? 

2. Explique o efeito de exec_3. 

 Variáveis e atribuição. Expressões 

Sub atrib_1() 
    Dim i As Integer 
    
    i = 2 
    MsgBox "Valor de i = " & i 
  
    i = 3 
    MsgBox "Valor de i = " & i 
  
    i = 3 * 3 + 8 
    MsgBox "Valor de i = " & i 
End Sub 
  
Sub atrib_2() 
    Dim i As Integer, n As Integer 
  
    i = 2    
    n = i 
    MsgBox "Valor de n = " & n 
     
    n = 2 * i + 1 
    MsgBox "Valor de n = " & n 
    n = n + 1 
    MsgBox "Valor de n = " & n 
     
    n = 2 * n + 8 
    MsgBox "Valor de n = " & n 
End Sub 
 

1. Explique o efeito das subrotinas atrib_1 e atrib_2. 

2. Suponha que n tem o valor 3 e i tem o valor 2. 
Qual o valor de n após cada um das seguintes instruções: 
a) n  =  2 *  ( i + 1) 
b) n = i ^2 
c) n = 10 * i / 4 
d) n = n + 1 
e) n = n ^2 
f) n = n * n 
g) n = n + i + 1 
h) n + 1 = i 

3. Sejam a e b duas variáveis. Qual a diferença entre a = b e b = a ? 

 Variáveis: Tipificação 

Sub tipo_1() 
    Dim i As Integer 
    i = 10 
    MsgBox "i contém -> " & i 
     
    Dim str As String 
    str = "Pim." 
    MsgBox "str contém -> " & str 
   
    Dim d As Date 
    d = Date + 1 
    MsgBox "d contém -> " & d 
  
    Dim dec As Double 
    dec = i / 3 
    MsgBox "dec contém -> " & dec 
  
    Dim tf As Boolean 
    tf = 1 = 2 
    MsgBox "tf contém -> " & tf 
End Sub 
 
Sub tipo_3() 
    Dim i As Integer 
    i = 13 
    i = i / 2 
    i = i * 2 
    MsgBox "i contém -> " & i 
End Sub 

  
Sub tipo_5() 
    Dim i As Integer 
    Dim str As String 
    Dim var As Variant 
     
    MsgBox TypeName(i) 
    MsgBox TypeName(str) 
    MsgBox TypeName(var) 
     
    var = 10 
    MsgBox "var com " & var & " fica " & 
TypeName(var) 
     
    var = "Pim" 
    MsgBox "var com " & var & " fica " & 
TypeName(var) 
  
    var = var = "Pim" 
    MsgBox "var com " & var & " fica " & 
TypeName(var) 
  
End Sub 
 

 

1. Explique o efeito da subrotina tipo_1. 

2. Qual o significado de cada um dos dois sinais = na instrução tf = 1 = 2 ? 

3. Explique o efeito da subrotina tipo_3 

4. Qual poderá ser o efeito de cada um das seguintes instruções (com as mesmas variáveis da subrotina 
tipo_1): 
a) i = "Olá" 
b) i = 3.33 
c) str = 33 

5. Explique o efeito de tipo_5. 



 

INFORMÁTICA DE GESTÃO 

 

 

© Informática de Gestão ISCTE-IUL / DCTI, 2011 6 

 Leitura com InputBox 

Sub input_1() 
    Dim n As Integer 
    n = InputBox("Diga um número: ") 
  
    MsgBox "O número que disse é o " & n 
    MsgBox "E o seguinte é o " & n + 1 
End Sub 
  
  
Sub input_2() 
    Dim nome As String 
     
    nome = InputBox("Diga o seu nome: ") 
    MsgBox "O seu nome tem " & Len(nome) & " letras." 
End Sub 
 

1. Explique o efeito das subrotinas input_1 e input_2. 

2. Faça uma subrotina que leia dois números e apresente a respectiva média (ie que faça aparecer uma 
inputbox a pedir um número; depois outra inputbox a pedir outro número; depois mostre a média numa 
janela msgbox). 

3. Faça uma subrotina que leia um número n e mostre o valor de 2n e o de n2. 

4. Faça uma subrotina que leia um número n e apresente o próximo múltiplo de 7 maior ou igual a n. 

 Acesso a células 

Sub cel_1() 
    Range("A1").Value = 1 
    Range("A2").Value = 2 
    Range("A3").Formula = "=A1+A2" 
End Sub 
 
Sub cel_2() 
    MsgBox Range("A3").Value 
    MsgBox Range("A3").Formula 
End Sub 
 
Sub cel_3() 
    Dim x As Variant 
    x = Range("B1").Value 
    MsgBox "A célula B2 contém o valor " & x 
End Sub 
 
Sub cel_5() 
    Dim x As Variant 
     
    x = InputBox("Valor: ") 
    Range("A1").Value = x 
End Sub 
 
Sub cel_10() 
    Dim s 
     
    s = InputBox("Célula: ") 
    MsgBox Range(s).Formula 
End Sub 

1. Explique o funcionamento das subrotinas cel_1 e cel_2. 

2. Explique o funcionamento da subrotina cel_3. 
Qual a razão para usar uma variável declarada como Variant ? 

3. Preencha as células B5 e B6 com dois números. Faça uma subrotina que apresente uma janela com a 
soma dos valores existentes nestas duas células. 

4. Explique o efeito da subrotina cel_5. 

5. Faça uma subrotina que leia dois valores e preencha as células B1 e B2 com o primeiro deles e as 
células F3 e F4 com o segundo. 

6. Na folha de cálculo, insira a fórmula B5:=FACT(B4). 
Faça uma subrotina que leia um número e apresente o respectivo factorial. 

7. Faça uma subrotina que leia duas notas e indique a nota final de um aluno. A nota final é a média das 
duas notas. Para ficar Aprovado o aluno tem que ter pelo 10 de média e nota mínina 7 em ambas as 
provas. 

8. É dada uma tabela com o número e nome dos alunos de uma turma. Faça uma subrotina que, dado um 
número, mostre o respectivo nome. 

9. Explique o funcionamento da subrotina cel_10. 

 Célula activa 

Sub act_1() 
    ActiveCell.Value = 77 
End Sub 
  
  
Sub act_2() 
  
    MsgBox "A célula activa é " & ActiveCell.Address & _ 
            " e contém " & ActiveCell.Formula 
End Sub 
  
  
Sub act_3() 
    Range("B3").Activate 
End Sub 
  
  
Sub act_4() 
  
    Dim s 
     
    s = InputBox("Célula: ") 
    Range(s).Activate 
End Sub 
  
Sub act_5a() 
    Range("F5").Value = "Viva eu." 
End Sub 
  
  
Sub act_5b() 
    Range("F5").Activate 
    ActiveCell.Value = "Viva eu." 
End Sub 
 

1. Explique o funcionamento das subrotinas act_1, act_2, act_3 e act_4. 

2. Pretende-se preencher a célula F5. 

3. Compare as duas alternativas act_5a e act_5b. 

 Execução condicional: If 

Sub if_1() 
    Dim n As Integer 
     
    n = InputBox("Número: ") 
    If n > 7 Then 
        MsgBox ("xxx") 
    End If 
    MsgBox ("zzz") 
End Sub 
  
 

Sub if_3() 
    Dim a As Integer, b As Integer 
     
    a = InputBox("Primeiro: ") 
    b = InputBox("Segundo: ") 
   
    If a > b Then 
        MsgBox "Resultado: " & a 
    Else 
        MsgBox "Resultado: " & b 
    End If 
End Sub 

1. Explique o efeito da subrotina if_1 quando para valores dados 10, 3 e 7. 

2. Complete a seguinte subrotina para obter o mesmo efeito que if_1 
 
Sub if_2() 
    Dim n As Integer 
    n = InputBox("Número: ") 
    If n ... 7 Then 
        MsgBox ("zzz") 
    else 
        ... 
    End If 
End Sub 

3. Explique o efeito da subrotina if_3. 

4. Complete a seguinte subrotina para obter o mesmo efeito  
 
Sub if_4() 
    Dim n, m As Integer 
    m = InputBox("Primeiro: ") 
    n = InputBox("Segundo: ") 
    If n > m Then 
        ... 
    End If 
    MsgBox "Resultado: " & m 
End Sub  

5. Faça uma subrotina if_5 que leia dois números e os apresente por ordem crescente. 

6. Complete a seguinte subrotina para obter o mesmo efeito 
 
Sub if_6() 
    Dim a, b, x As Integer 
    a = InputBox("Primeiro: ") 
    b = InputBox("Segundo: ") 
    If b > a Then 
        ... 
    End If 
    MsgBox "Por ordem: " & a & " - " & b 
End Sub 



 

INFORMÁTICA DE GESTÃO 

 

 

© Informática de Gestão ISCTE-IUL / DCTI, 2011 10 

 Execução condicional: If (II) 

Sub if_7() 
  
    Dim n As Integer, resto As Integer 
     
    n = InputBox("Número: ") 
    resto = n Mod 2 
    If resto = 0 Then 
        MsgBox n & " é par." 
    Else 
        MsgBox n & " é ímpar." 
    End If 
     
End Sub 

1.  Descreva o efeito da subrotina if_7. 

2. Complete a subrotina if_8 para obter o mesmo efeito. 
 
Sub if_8() 
    Dim n As Integer, resto As Integer 
    Dim msg As String      
    n = InputBox("Número: ") 
    msg = n & " é ímpar." 
    If n Mod 2 = 0 Then 
        ... 
    End If 
    MsgBox msg 
End Sub 

3. Complete a subrotina if_9 para obter o mesmo efeito. 
 
Sub if_9() 
    Dim n As Integer, resto As Integer 
    Dim msg As String 
    n = InputBox("Número: ") 
    msg = n & " é" 
    If n Mod 2 = 0 Then 
        ... 
    Else 
        ... 
    End If 
    MsgBox msg 
End Sub 
 

4. Considere duas células B3 e C3. Faça uma subrotina que leia um número e o adicione àquela que, no 
momento, contiver um número menor; e, depois, indique qual delas ficou com o número maior. 

 Condicional: Várias hipóteses. 

Sub ifs_1() 
    Dim a As Integer, b As Integer 
    a = InputBox("Primeiro: ") 
    b = InputBox("Segundo: ") 
 
    If a > b Then 
        MsgBox "Primeiro ganha." 
    Else 
        If a < b Then 
            MsgBox "Segundo ganha." 
        Else 
            MsgBox "Empatam." 
        End If 
    End If 
End Sub 

Sub ifs_2() 
    Dim a As Integer, b As Integer 
    a = InputBox("Primeiro: ") 
    b = InputBox("Segundo: ") 
  
    If a > b Then 
        MsgBox "Primeiro ganha." 
    ElseIf a < b Then 
            MsgBox "Segundo ganha." 
    Else 
            MsgBox "Empatam." 
    End If 
End Sub 
 

 
Sub ifs_3() 
    Dim a As Integer, b As Integer 
    a = InputBox("Primeiro: ") 
    b = InputBox("Segundo: ") 
     
    If a > b Then MsgBox "Primeiro ganha." 
    If a < b Then MsgBox "Segundo ganha." 
    If a = b Then MsgBox "Empatam." 
End Sub 
 

Sub ifs_4() 
    Dim n As Integer 
    n = InputBox("Numero: ") 
  
    Select Case n 
    Case 10 
        MsgBox ("Situação A") 
    Case 11 
        MsgBox ("Situação B") 
    Case Is < 10 
        MsgBox ("Situação C") 
    Case Is = 12 Or n = 13 
        MsgBox ("Mau aspecto!") 
    Case Else 
        MsgBox "Nenhuma das anteriores" 
    End Select     
End Sub 
 

1. Explique e compare o funcionamento das subrotinas ifs_1, ifs_2 e ifs_3. 

2. Considere a mesma situação do exercício 10 do ponto anterior. Faça uma subrotina que leia um número 
(1, 2, 3 ou 4) e, consoante esse número, indique: 
a) o valor de B3 
b) o valor de C3 
c) a diferença (em valor absoluto) 
d) a soma 

3. Explique o funcionamento da subrotina ifs_5. 

4. Refaça  subrotina do ponto 4, desta vez usando Select. 

 Expressões lógicas 

Sub and_1() 
    Dim n As Integer 
    n = InputBox("Numero entre 1 e 13 : ") 
    If n >= 10 And n <= 13 Then 
        MsgBox "Certo." 
    Else 
        MsgBox "Errado." 
    End If 
End Sub  

1. Explique o funcionamento da subrotina and_1. 

2. Substitua a condição a>=10 and a <=13 por ... or ... . 

 Repetição: For. 

Sub for_1() 
    Dim i as Integer 
    MsgBox "Inicio" 
    For i = 1 To 3 
        MsgBox "Um piano." 
    Next i 
    MsgBox "Fim." 
End Sub 
Sub for_2() 
    Dim i as Integer 
    For i = 1 To 3 
        MsgBox "Piano nº " & i 
    Next i 
End Sub 
Sub for_3() 
    Dim i as Integer 
    For i = 10 To 16 Step 2 
        MsgBox "Piano nº " & i / 2 
    Next i 
End Sub 
Sub for_6() 
    Dim i As Integer, n As Integer 
     
    For i = 1 To 5 
        n = 2 * i 
    Next 
    MsgBox "Final: i= " & i & " e n= " & n 
End Sub 
Sub for_7() 
    Dim n As Integer, i As Integer 
    n = InputBox("Quantas vezes ? ") 
    For i = 0 To n - 1 
        MsgBox ("Pim.") 
    Next 
End Sub 

Sub for_8() 
    Dim i As Integer, n As Integer 
    For i = 1 To 3 
        n = InputBox("Numero: ") 
        MsgBox n 
    Next 
End Sub 
 
Sub for_9() 
    Dim i As Integer, n As Integer 
    For i = 1 To 3 
        n = InputBox("Numero: ") 
        If n Mod 2 = 0 Then 
            MsgBox "Par." 
        Else 
            MsgBox "É ímpar." 
        End If 
    Next 
End Sub 
 
Sub for_10() 
    Dim i As Integer, n As Integer 
    For i = 1 To 3 
        n = InputBox("Numero: ") 
        If n Mod 2 = 0 Then 
            MsgBox n 
        End If 
    Next 
End Sub 

1. Explique as subrotinas for_1, for_2 e for_3. 

2. Qual a lista de valores de n gerados nos seguintes ciclos: 
a) for n = 1 to 7 step 2  
b) for n = 7 to 1 
c) for n = 7 to 6 step –1 

3. Complete, para obter 5 repetições: 
a) for i = 0 to ... 
b) for i = -1 to ... 
c) for i = ... to 5 step –1 

4. Quais os "valores finais " da subrotina for_6 ? 

5. Quantas vezes aparace a janela "Pim." na execução da subrotina for_7 ? 

6. Explique as subrotinas for_8, for_9 e for_10 

7. Faça uma subrotina que leia 5 números e faça aparecer uma janela (MsgBox) com a mensagem "Que 
grande." de cada vez que um deles for maior que 1000. 

 Selection 

Sub sel() 
    MsgBox "Inicio." 
    Range("A2:B5").Select 
    MsgBox Selection.Address 
     
    Selection.Value = "33" 
    MsgBox "Fim." 
End Sub 

1. Explique a subrotina sel. 

2. Suponhamos que foi efectuada a instrução: Range(F2:G4).Select. 
Qual a diferença entre Range(F2:G4) = 77 e Selection.Value = 77 

3. Faça uma subrotina que leia um valor e preencha todas as células B1:B5 com esse mesmo valor. 

4. Compare as duas  
a) Range("F5:G9").Select 
b) Range("F5:G9").Activate 

5. Compare e indique quais as opções de menu correspondentes a cada uma das seguintes instruções? 
a) Selection.Clear 
b) Selection.ClearContents 
c) Selection.ClearFormats 

 Range: algumas propriedades e métodos 

Sub prop1() 
    Range("A2:B5").Clear 
End Sub 
 
Sub prop2() 
    MsgBox ActiveCell.NumberFormat 
End Sub 
 
Sub prop3() 
    Range("F5:F9").FillDown 
End Sub 
 
Sub prop4() 
    ActiveCell.Copy Range("A1") 
End Sub 
 
Sub prop5() 
    Selection.PrintPreview 
End Sub 

1. Explique o efeito das subrotinas prop1 e prop2? 

2. Preencha a célula F5 e apague o conteúdo das 4 células abaixo dela. 
a) Nestas condições, qual o efeito da subrotina prop3? 
b) Em geral, qual o efeito de Selection.FillDown 

3. Qual o efeito de subrotina prop4? 

4. O que faz cada uma das seguintes instruções? 
a) Range("A1:B2").Copy Range("F5") 
b) Selection.Copy Range("F5") 
c) Selection.Copy ActivelCell 

 Objecto Range 

Sub rang1() 

    Dim r As Range 
     
    Set r = Range("B3") 
    r.Value = "Passo 1" 
    MsgBox r.Address 
     
    Set r = Selection 
    r.Value = "Passo 2" 
    MsgBox r.Address 
  
    Set r = ActiveCell 
    r.Value = "Passo 3" 
    MsgBox r.Address 
End Sub 

1. Explique o fucionamento da subrotina rang1. 

2. Qual o efeito de 
a) set r = ActiveCell 

r.Clear 
b) set r = Selection 

r.Clear 
c) set r = Selection 

r.PrintPreview 
d) set r = Range("F5:F9") 

r.Select 
e) set r = Range("F5:F9") 

r.Activate 

3. Qual o efeito de r.Copy d após 
a) set r = ActiveCell 

set d = Range("F5") 
b) set r = Range ("A2:F5") 

set d = Range("F6") 
c) set r = Selection 

set d = Range("A1") 

 Range: Count 

Sub count_1() 

    MsgBox Range("A1:B2").Count 
End Sub 
  
  
Sub count_2() 

    Dim r As Range 
    Dim n As Integer 
     
    Set r = Range("A1:B2") 
    MsgBox r.Count 
    ' 
    Set r = Selection 
    MsgBox r.Count 
    ' 
    Set r = ActiveCell 
    MsgBox r.Count 
End Sub 
  
Sub count_3() 

    Dim n As Integer 
    Dim s As String 
     
    n = Selection.Count 
    If n = 1 Then 
        MsgBox "Está seleccionada 1 célula" 
    Else 
        MsgBox "Estão seleccionadas " & n & " células." 
    End If 
  
End Sub 
  
Sub count_4() 

    Dim r As Range 
    Set r = Range("B5:D12") 
    MsgBox r.Rows.Count 
End Sub 
 

1. Explique as subrotinas count_1, count_2, count_3? 
Qual o resultado de ActiveCell.Count ? 

2. Explique a subrotina count_4 ? 
Em que condições Selection.Count e Selection.Columns.Count dão o mesmo valor ? 

 Range: count e cells 

Sub cells_1() 
    Dim r As Range, c As Range 
  
    Set r = Range("B5:B10") 
     
    Set c = r.Cells(1) 
    MsgBox c.Address 
     
    Set c = r.Cells(3) 
    MsgBox c.Address 
  
    Set c = r.Cells(6) 
    MsgBox c.Address 
  
    Set c = r.Cells(9) 
    MsgBox c.Address 
End Sub 
 
Sub cells_2() 
    ActiveCell.Cells(2).Activate 
End Sub 
 
Sub cells_3() 
    Dim r As Range 
    Set r = ActiveCell.Cells(2) 
    r.Value = ActiveCell.Value + 1 
End Sub 

Sub cells_4() 
    ActiveCell.Copy ActiveCell.Cells(2) 
End Sub  
 
Sub cells_5() 
    Dim r As Range, c As Range 
 
    Set r = Range("B5:C8") 
    r.Select 
     
    Set c = r.Cells(1, 1) 
    c.Activate 
    MsgBox c.Address 
     
    Set c = r.Cells(1, 2) 
    c.Activate 
    MsgBox c.Address 
 
    Set c = r.Cells(2, 2) 
    c.Activate 
    MsgBox c.Address 
 
    Set c = r.Cells(3, 4) 
    c.Activate 
    MsgBox c.Address 
End Sub 

1. Descreva o efeito de cells_1. 
a) Experimente a mesma subrotina com outros conjuntos de células, por exemplo: 

Range("B5:C8"), Selection, ActiveCell. 
b) Em geral, qual o efeito de range.cells(i)? 

2. Qual o efeito de cells_2 ? 

3. Qual o efeito de cada uma destas instruções: 
a) ActiveCell.Cells(2).Value = "Pim" 
b) Msgbox ActiveCell.Cells(2).Value 
c) ActiveCell = ActiveCell.Cells(2).Value 
d) ActiveCell.Cells(2).Value = ActiveCell.Value & "X" 
e) MsgBox Selection.Cells(2).Count 

4. Faça uma subrotina que escreva "Ola" na célula abaixo da célula activa. 

5. Explique as subrotinas cells_3 e cells_4? 

6. Faça uma subrotina que copie a célula activa para as três células abaixo dela. 

7. Faça um subrotina que leia um número e preencha com ele a célula activa e as três células abaixo dela. 

8. Explique a subrotina cell_5. Qual o efeito de range.cells(i,j)? 

 Range: count e cells 

Sub acc_1() 
    Dim i As Integer, n As Integer 
    Dim r As Range 
     
    Set r = Range("F5:F9") 
    For i = 1 To r.Count 
        MsgBox r.Cells(i).Address 
        r.Cells(i).Value = Time 
    Next 
End Sub 

Sub acc_2() 
    Dim i As Integer, n As Integer 
    Dim r As Range 
     
    Set r = Selection 
    For i = 1 To r.Count 
        r.Cells(i).Value = i 
    Next 
End Sub 

1. Explique o funcionamento das subrotina acc_1 e acc_2. 

2. Qual seria o efeito de, em acc_2, modificar r.Cells(i).Value = i para 
a) r.Cells(i).Value = 100 + i 
b) r.Cells(i).Value = 100 + i 
c) r.Cells(i).Value = r.Count – i 
d) r.Cells(r.Count-i).Value = i 

3. Faça uma subrotina que preencha a Selection com números pares (começando em 2). 

 Range: acesso com For Each 

Sub feach_1() 
    Dim r As Range, c As Range 
     
    Set r = Range("G5:G9") 
    For Each c In r 
        MsgBox c.Address 
        c.Value = Time 
    Next 
End Sub 
  
  
Sub feach_2() 
    Dim r As Range, c As Range 
    Dim d As Date 
     
    d = Date 
    Set r = Range("H5:H9") 
    For Each c In r 
        c.Value = d 
        d = d + 1 
    Next 
  
End Sub 

1. Explique o funcionamento das subrotina feach_1 e feach_2. 

2. Faça uma subrotina que preencha as células do grupo J5:J9 com número de 10 a 15. 

3. Faça uma subrotina que leia um número e preencha as células do grupo K5:K9 com os números 
seguintes ao número dado. 
a)  

 Acumulador 

Sub acum_1() 
    Dim r As Range, c As Range 
    Dim soma 
     
    Set r = Range("B2:B7") 
    r.Select 
    soma = 0 
    For Each c In r 
        soma = soma + c.Value 
        c.Activate 
        MsgBox "Até aqui dá " & soma 
    Next 
    MsgBox "Soma = " & soma 
End Sub 
  
Sub acum_2() 
    Dim r As Range, c As Range 
    Dim soma 
     
    Set r = Range("B2:B7") 
    r.Select 
    soma = 0 
    For Each c In r 
        soma = soma + c.Value 
        c.Cells(1, 2).Value = soma 
    Next 
    MsgBox "Soma = " & soma 
End Sub 
 

Preencha as células B2:B7 com números. 

1. A subrotina acum_1 calcula a soma dos valores contidos nas células do grupo B2:B7. 
Explique o funcionamento da subrotina, em particular o mecanismo de acumulação na variável soma. 

2. Antes de experimentar, tente prever qual o efeito da subrotina acum_2. A única diferença é trocar 
c.Activate 
MsgBox "Até aqui dá " & soma 
por 
c.Cells(1, 2).Value = soma 

3. Faça uma subrotina que calcule a média dos valores das células do grupo B2:B7. 

4. Faça uma subrotina que leia 5 números, insira cada um deles numa célula e a soma noutra célula. 



 

INFORMÁTICA DE GESTÃO 

 

 

© Informática de Gestão ISCTE-IUL / DCTI, 2011 22 

 Contador 

Sub conta_1() 
    Dim r As Range, c As Range 
    Dim n As Integer 
   
    Set r = Range("B2:B7") 
    r.Select 
     
    n = 0 
    For Each c In r 
        c.Activate 
        If c.Value Mod 2 = 0 Then 
            n = n + 1 
            MsgBox "Este conta. Passa a " 
& n 
        Else 
            MsgBox "Este não conta" 
        End If 
    Next 
    MsgBox "Ao todo são " & n 
End Sub 

Sub conta_2() 
    Dim c As Range 
    Dim n As Integer 
     
    n = 0 
    For Each c In Selection 
        n = n + 1 
    Next 
    MsgBox "Resultado: " & n 
End Sub 
  
Sub conta_3() 
    Dim r As Range, c As Range 
    Dim n As Integer 
   
    Set r = Range("B2:B7") 
    r.Select 
     
    n = 0 
    For Each c In r 
        If c.Value = "" Then 
            n = n + 1 
        End If 
    Next 
    MsgBox "Ao todo são " & n 
End Sub 

1. A subrotina conta_1 pretende contar o número de valores pares das células do grupo B2:B7. Explique o 
funcionamento da subrotina, em particular o mecanismo de contagem suportado na variável n. 

2. Qual o resultado da subrotina conta_2 ? 
De que outra forma, mais simples, se pode obter o mesmo resultado ? 
Apague o conteúdo de uma ou mais das células do grupo B2:B7. 
As outras continuam a ter números. 

3. Qual o resultado da subrotina conta_3 ? 

4. Faça uma subrotina que calcule a média doa valores de B2:B7, não contabilizando as células vazias (ie, 
como o AVERAGE(B2:B7) faria nas mesmas circunstâncias. 

5. Quais das seguintes expressões servem para ver se a célula activa está vazia: 
a) ActiveCell.Value = "" 
b) Len(ActiveCell.Value) = 0 
c) ActiveCell.Value = 0 
d) IsEmpty(ActiveCell.Value) 

6. Preencha a célula D2 com um número. 
Faça uma subrotina que preencha as células D3 até D9 com os números seguintes a D2. 

7. Preencha as células E2 e E3 com dois números. 
Faça uma subrotina que preencha as células até E9 com termos seguintes da série iniciada com os 
valores de E2 e E3. 

8. Faça uma subrotina que leia um número e indique quantas vezes ele aparece nas células B2:B7. 

 Detecção. 

Sub det_1() 
    Dim r As Range, c As Range 
    Dim i As Integer 
    Dim max As Integer 
   
    Set r = Range("K2:K9") 
    r.Select  
     
    'a célula activa é a primeira do grupo, ou seja K2 
    max = ActiveCell.Value 
    MsgBox "Começamos com Max = " & max 
     
    'começamos a comparar a partir da 2ª célula do 
grupo 
    For i = 2 To r.Count 
        Set c = r.Cells(i) 
        c.Activate 
        If c.Value > max Then 
            max = c.Value 
            MsgBox "Trocamos Max para " & max 
        Else 
            MsgBox "Max já é maior/igual a este. Fica 
na mesma" 
        End If 
    Next 
    MsgBox "No fim Max= " & max 
End Sub 
 

Sub det_3() 
    Dim r As Range, c As Range 
    Dim i As Integer 
    Dim max As Integer 
   
    Set r = Range("K2:K9") 
    r.Select 
     
    max = ActiveCell.Value 
    For Each c In r 
        If c.Value > max Then 
            max = c.Value 
        End If 
    Next 
    MsgBox "Max -> " & max 
End Sub 

 

Preencha as células do grupo K2:K9 com números  

1. A subrotina det_1 calcula o maior valor existente nas células do grupo K2:K9 (mais ou menos o que 
faz a função MAX do Excel). Explique o funcionamento, em particular o mecanismo de deteção do 
maior suportado na variável max. 

2. O mecanismo geral é muito parecido com o da soma e o da contagem. Só que, aqui, inicia-se a variável 
max com o valor da primeira célula, em vez do valor 0. Isso obriga-nos a começar só na segunda célula 
e portanto a usar o For...Next em vez do For Each que seria muito mais simples. Não bastaria, 
simplesmente, começar com max = 0 ? 

3. Verifique a versão det_3 muito semelhante a det_1. 

4. De que forma é tratada a primeira célula (neste caso K2) ? 

5. Faça uma subrotina que calcule o menor valor contido nas células do grupo K2:K9. 

6. Faça uma subrotina que leia 5 números, insira cada um deles numa célula e o maior noutra. 

7. Faça uma subrotina que calcule o maior número existente em K2:K9 e indique, também, quantas vezes 
esse número aparece. 

8. Faça uma subrotina que calcule a média dos valores de K2:K9 e indique quantos dos valores estão 
acima dessa média e quantos estão abaixo. 

 Localização e flags. 

Sub flag_1() 
    Dim r As Range, c As Range 
    Dim n As Integer 
    Dim sn As String 
   
    Set r = Range("K2:K9") 
    r.Select 
      
    n = InputBox("Número a procurar ? ") 
        
    For Each c In r 
        c.Activate 
        If c.Value = n Then 
            sn = "Sim, existe (ó ele 
aqui!)." 
            MsgBox sn 
        Else 
            sn = "Não, não existe." 
            MsgBox sn 
        End If 
    Next 
    MsgBox "Conclusão: " & sn 
End Sub 
 
Sub flag_2() 
    Dim r As Range, c As Range 
    Dim n As Integer 
    Dim sn As Boolean 
   
    Set r = Range("K2:K9") 
    r.Select 
     
    n = InputBox("Número a procurar ? ") 
        
    sn = False 
    For Each c In r 
        c.Activate 
        If c.Value = n Then 
            sn = True 
            MsgBox "Aqui concluí que 
existe sim senhor." 
        End If 
    Next 
    If sn Then 
        MsgBox "Conclusão: Existe" 
    Else 
        MsgBox "Conclusão: Não existe" 
    End If 
End Sub 

 
Sub flag_3() 
    Dim r As Range, c As Range 
    Dim n As Integer 
    Dim i As Integer, idx As Integer 
   
    Set r = Range("K2:K9") 
    r.Select 
     
    n = InputBox("Número a procurar ? ") 
        
    idx = 0 
    For i = 1 To r.Count 
        Set c = r.Cells(i) 
        c.Activate 
        If c.Value = n Then 
            idx = i 
        End If 
    Next 
    If i = 0 Then 
        MsgBox "Não existe." 
    Else 
        r.Cells(idx).Activate 
    End If 
End Sub 
 

1. A subrotina flag_1 lê um valor e diz se ele se encontra ou não nas células do grupo K2:K9. 
Mas, infelizmente, não funciona. Explique porquê. 

2. Considere a versão flag_2 do mesmo problema. Explique o funcionamento desta versão, em particular 
da variável de sinalização sn. 

3. Qual o efeito da subrotina flag_3 ? 

 Localização e Exit. 

Sub exit_1() 
    MsgBox "Início" 
    Exit Sub 
    MsgBox "Fim" 
End Sub 
  
  
Sub exit_2() 
    Dim n As Integer, i As Integer 
    Dim msg As String 
     
    For i = 1 To 5 
        n = InputBox("Diga o " & i & "º 
valor ") 
        If n = 0 Then 
            MsgBox "Saída." 
            Exit Sub 
        End If 
    Next 
    MsgBox "Fim." 
End Sub 

Sub exit_4() 
    Dim r As Range, c As Range 
    Dim n As Integer, i As Integer 
   
    Set r = Range("K2:K9") 
    r.Select 
     
    n = InputBox("Número a procurar ? ") 
        
    For i = 1 To r.Count 
        Set c = r.Cells(i) 
        If c.Value = n Then 
            c.Activate 
            Exit For 
        End If 
    Next 
    If i > r.Count Then 
        MsgBox "Não existe." 
    End If 
End Sub 

 

1. Explique o funcionamento da subrotina exit_1 

2. De que formas pode terminar a subrotina exit_2 ? 

3. Na subrotina anterior substitua Exit Sub por Exit For. Qual o efeito ? 

4. Explique o funcionamento da subrotina exit_4. 

5. Indique uma situação em que flag_3 e exit_4 têm efeitos não exactamente iguais. 

 Ciclos de condição 

Sub while_1() 
    Dim i As Integer 
     
    i = 1 
    Do While i < 5 
        MsgBox "Janela nº " & i 
        i = i + 1 
    Loop 
    MsgBox "O valor final do i é " & i 
End Sub 
 
 
Sub while_2() 
    Dim i As Integer 
     
    i = 1 
    Do Until i = 5 
        MsgBox "Janela nº " & i 
        i = i + 1 
    Loop 
    MsgBox "O valor final do i é " & i 
End Sub 
 
Sub while_3() 
    Dim i As Integer 
     
    i = 1 
    Do 
        MsgBox "Janela nº " & i 
        i = i + 1 
    Loop Until i = 5 
    MsgBox "O valor final do i é " & i 
End Sub 
  
  
  
 
 

Sub while_4() 
    Dim i As Integer 
     
    i = 1 
    Do 
        MsgBox "Janela nº " & i 
        i = i + 1 
    Loop While i < 5 
    MsgBox "O valor final do i é " & i 
End Sub 
 
 
Sub while_5() 
    Dim i As Integer 
    Dim n As Integer, soma As Integer 
     
    i = 0 
    soma = 0 
    Do 
        n = InputBox("Numero: ") 
        soma = soma + n 
        i = i + 1 
    Loop Until n = 0 
    MsgBox "São " & i & " números e a soma 
é " & soma 
End Sub 
 
  
  
Sub while_7() 
    Dim c As Range 
     
    Set c = ActiveCell 
    Do Until c.Value = "" 
        MsgBox "Avança..." 
        Set c = c.Cells(2) 
        c.Activate 
    Loop 
    MsgBox "Parou." 
End Sub 

 

1. Explique o efeito das subrotinas while_1, while_2, while_3 e while_4. 

2. Qual o resultado da subrotina while_5 ? 

3. Refaça a subrotina while_5 usando cada um das outras formas do Do...Loop. 

4. Active uma célula não vazia, por exemplo K2. 

5. Qual o efeito da subrotina while_7 ? 

 4. Range: offset, current region, resize 

Sub off_1() 
    Dim r As Range, c As Range 
  
    Set r = Range("B2") 
  
    Set c = r.Offset(0, 1) 
    c.Activate 
    MsgBox "B2 com offset 0, 1" 
     
    Set c = r.Offset(1, 0) 
    c.Activate 
    MsgBox "B2 com offset 1, 0" 
  
    Set c = r.Offset(4, 2) 
    c.Activate 
    MsgBox "B2 com offset 4, 2" 
End Sub 
  
Sub off_2() 
    Dim r As Range, c As Range 
  
    Set r = Range("B2:F9") 
    r.Select 
    MsgBox "Incial..." 
     
    Set r = r.Offset(4, 2) 
    r.Select 
    MsgBox "Depois de offset 4,2" 
End Sub 
 
Sub off_3() 
    Dim r As Range 
    Set r = Range("B2") 
    r.Select 
    MsgBox "Incial..." 
     
    Set r = r.CurrentRegion 
    r.Select 
    MsgBox "Depois de CurrentRegion" 
End Sub 
 
Sub off_4() 
    Selection.CurrentRegion.Select 
End Sub 
 

Sub off_5() 
    Dim r As Range, c As Range 
     
    Set r = Range("B2") 
    r.Select 
    MsgBox "Inicial..." 
  
  
    Set c = r.Resize(3, 1) 
    c.Select 
    MsgBox "Com 3 linhas e uma coluna" 
  
    Set c = r.Resize(1, 3) 
    c.Select 
    MsgBox "Com 1 linha e 3 colunas" 
  
    Set c = r.Resize(4, 2) 
    c.Select 
    MsgBox "Com 4 linhas e 2 colunas" 
End Sub 
  
Sub off_6() 
    Dim r As Range, c As Range 
     
    Set r = Range("B2:F9") 
    r.Select 
    MsgBox "Inicial..." 
     
    Set c = r.Resize(, 1) 
    c.Select 
    MsgBox "Com uma coluna" 
     
    Set c = r.Resize(1) 
    c.Select 
    MsgBox "Com uma linha" 
End Sub 
 
 

Preencha o grupo B2:F10 com uma tabela dado com as seguintes colunas (representando uma pauta): 
Número, Nome, T1, T2, Final. A nota Final é a média, arredondada, de T1 e T2 

1. Explique o funcionamento das subrotinas off_1, off_2. 

2. Explique o funcionamento de off_3. 

3. Experimente off_4 com diferentes Selection. 
Em geral, qual o efeito de CurrentRegion ?  

4. Explique o efeito de off_5 e off_6. 

 4. Range: offset, current region, resize - II 

Sub off_7() 
    Dim r As Range 
     
    Set r = Range("B2") 
    r.Select 
    MsgBox "Passo 1" 
     
    Set r = r.CurrentRegion 
    r.Select 
    MsgBox "Passo 2" 
     
    Set r = r.Offset(0, 2) 
    r.Select 
    MsgBox "Passo 3" 
     
    Set r = r.Resize(, 1) 
    r.Select 
    MsgBox "E pronto." 
End Sub 
 
Sub off_8() 
    Dim r As Range 
    Dim n As Integer 
     
    Set r = Range("B2") 
    r.Select 
    MsgBox "Passo 1" 
     
    Set r = r.CurrentRegion 
    r.Select 
    MsgBox "Passo 2" 
     
    n = r.Rows.Count 
    Set r = r.Offset(n, 0) 
    r.Select 
    MsgBox "Passo 3" 
     
    Set r = r.Resize(1, 1) 
    r.Select 
    MsgBox "E pronto." 
End Sub 

Sub off_9() 
    Dim r As Range 
     
    Set r = Range("B2").CurrentRegion 
    Set r = r.Offset(r.Rows.Count, 0) 
    Set r = r.Resize(1, 1) 
    ' 
    r.Value = r.Offset(-1, 0).Value + 1 
    r.Offset(0, 1).Value = "Zé" 
    r.Offset(0, 2).Value = 10 
    r.Offset(0, 3).Value = 12 
    ' 
    Set r = r.Offset(0, 4) 
    r.Offset(-1, 0).Copy r 
End Sub 
  
Sub off_11() 
    Dim r As Range 
     
    Set r = Range("B2").CurrentRegion 
    Set r = r.Resize(, 1) 
     
    Dim n As Integer 
    n = InputBox("Numero do aluno: ") 
     
    Dim c As Range 
    For Each c In r 
        If c.Value = n Then 
            c.Activate 
            Exit For 
        End If 
    Next 
End Sub 
 

1. A subrotina off_7 pretende seleccionar a coluna das notas T1. 
Explique a manobra efectuada. 

2. A subrotina off_8 pretende posicionar a célula activa na primeira célula em branco da coluna Número. 
Explique a manobra efectuada. 

3. A subrotina off_9 acrescenta um novo aluno à tabela. 

4. Explique de que forma aparecem o número do novo aluno e a fórmula de cálculo da média. 

5. Faça uma subrotina que leia os dados de um novo aluno (número, nome, nota T1 e nota T2) e 
acrescente o aluno à tabela. 

6. Explique o efeito da subrotina off_11. 

7. Faça uma subrotina que, dado o número de um aluno, mostre o respectivo nome. 

 Conteúdo das células 

Sub val_1() 
    MsgBox ActiveCell.Value 
    MsgBox ActiveCell.Value2 
    MsgBox ActiveCell.Formula 
End Sub 
  
  
  
Sub val_2() 
    MsgBox "Vazia ? " & IsEmpty(ActiveCell.Value) 
    MsgBox "Número ? " & IsNumeric(ActiveCell.Value) 
End Sub 
  
  
  
Sub val_3() 
    MsgBox "Vazia ? " & IsEmpty(ActiveCell.Value2) 
    MsgBox "Número ? " & IsNumeric(ActiveCell.Value2) 
End Sub 
  

1. Experimente a subrotinas val_1, val_2 e val_3 em células com diferentes valores e formatos. 

2. Quais os resultados de IsNumeric(ActivelCell.Value2) para: 
a) uma célula vazia 
b) uma célula contendo o valor 0 
c) uma célula contendo uma data 

Preencha o grupo H2:H9 com números ou texto, deixando também uma ou outra célula vazia. 

3. A função COUNTA do Excel conta todas as células com algum conteúdo (não vazias). 
Faça uma subrotina que apresente o mesmo resultado que COUNTA(H2:H9). 

4. A função COUNT do Excel conta as células que contêm números. Portanto, ignora as que têm texto e 
as vazias. Faça uma subrotina que apresente o mesmo resultado que COUNT(H2:H9). 

5. Faça uma subrotina que dê o mesmo resultado que AVERAGE(H2:H9) 

6. Faça uma subrotina que dê o mesmo resultado que MAX(H2:H9). 

 Utilização de funções da folha de cálculo 

Sub wks_1() 
    MsgBox WorksheetFunction.Pi() 
End Sub 
  
  
  
Sub wks_2() 
    Dim n As Integer 
    n = InputBox("Numero: ") 
    n = WorksheetFunction.Ceiling(n, 7) 
    MsgBox "Próximo múltiplo de 7 é " & n 
End Sub 
  
  
  
Sub wks_3() 
    Dim n1 As Integer, n2 As Integer 
    Dim media As Double 
  
    n1 = InputBox("Nota 1 :") 
    n2 = InputBox("Nota 2 :") 
    media = WorksheetFunction.Average(n1, n2) 
    MsgBox "Média: " & media 
End Sub 
  
  
  
Sub wks_4a() 
    MsgBox WorksheetFunction.Sum("A1:B2") 
End Sub 
  
 
  
Sub wks_4b() 
    Dim r As Range 
    Set r = Range("A1:B2") 
    MsgBox WorksheetFunction.Sum(r) 
End Sub 
  
  
Sub wks_5() 
    MsgBox WorksheetFunction.IsNumber(ActiveCell) 
End Sub 

1. Explique o efeito da subrotinas wks_1, wks_2 e wks_3. 

2. Qual das duas versões wks_4a ou wks_4b lhe parecem adequadas para obter a soma dos valores das 
células do grupo A1:B2 ? 

3. Experimente a subrotina wks_5 para diferentes células. 

4. Faça uma subrotina que dê o mesmo resultado que COUNT(H2:H9). 

 Encadeamento de subrotinas 

Sub sub1() 
    MsgBox "Início..." 
    exec_1      'subrotina exec_1 do Grupo 1 
    MsgBox "Continuamos..." 
    exec_3      'subrotina exec_3 do Grupo 2 
    MsgBox "E fim." 
End Sub 
  
  
  
Sub sub2() 
    MsgBox "Início de sub_2" 
    sub2a 
    exec_3 
    MsgBox "Fim de sub_2" 
End Sub 
  
  
  
Sub sub2a() 
    MsgBox "Início Sub_2a" 
    exec_3 
    MsgBox "Fim de sub_2a" 
End Sub 
  
  

1. Considere a subrotina sub1 (que chama as subrotinas exec_1 e exec_2 do Grupo 1). 
Explique a sequência de janelas que aparece. 

2. Antes de experimentar, preveja a sequência de janelas resultante da execução da subrotina sub2 (que, 
por sua vez, chama a subrotina sub2a). 

 Argumentos 

 
Sub arg1() 
    Dim i As Integer 
     
    arg1a (17) 
    i = 20 
    arg1a (i) 
    arg1a (i + 1) 
End Sub 
  
  
  
Sub arg1a(n As Integer) 
    MsgBox "Recebi n -> " & n 
End Sub 
  
  
  
Sub arg2() 
    Dim i As Integer, str As String 
     
    i = 1001 
    str = "Pimba" 
    arg2a 33, "Biba" 
    arg2a i, "Moi" 
    arg2a i + 1, str 
    arg2a i, "Moi " & str 
End Sub 
  
  
  
Sub arg2a(n As Integer, s As String) 
    MsgBox n & " , " & s 
End Sub 
  
  

1. Explique o efeito da subrotina arg1 (que, por sua vez, chama a subrotina arg1a). 

2. Explique o funcionamento da subrotina arg2 (que, por sua vez, chama arg2a). 

3. Quais das seguintes chamadas são aceitáveis: 
a) arg1a   99  ,  "Olá" 
b) arg2a   99  ,  99 
c) arg2a   99 
d) arg1a   "Olá" 

 Funções 

Function fex1() 
    fex1 = 3.14 
End Function 
  
Sub fex1b() 
    MsgBox "Resultado da função -> " & fex1() 
End Sub 
  
Sub fex2() 
    Dim a As Integer, b As Integer 
    Dim m As Integer 
    a = InputBox("Valor 1: ") 
    b = InputBox("Valor 1: ") 
    m = f_maior(a, b) 
    MsgBox "Maior : " & m 
End Sub 
  
Function f_maior(x As Integer, y As Integer) 
    If x > y Then 
        f_maior = x 
    Else 
        f_maior = y 
    End If 
End Function 
  
Sub fex3() 
    Dim a As Integer 
    a = InputBox("Valor: ") 
    a = f_ceiling(a, 7) 
    MsgBox "Resultado: " & a 
End Sub 
  
Function f_ceiling(x As Integer, m As Integer) 
    Dim resto As Integer 
    resto = x Mod m 
    If resto = 0 Then 
        f_ceiling = x 
    Else 
        f_ceiling = x - resto + m 
    End If 
End Function 

1. A função fex1 não tem argumentos e dá como resultado o número 3.14. 
Experimente a subrotina fex1b e confirme o resultado da função fex1. 

2. A função f_maior ( a, b ) recebe dois argumentos, ambos números inteiros, e dá como resultado o 
maior deles. Explique o efeito da subrotina fex2 que usa essa função. 

3. A função f_ceiling(a,m) é parecida com a função CEILING do Excel. 
Explique o efeito da subrotina fex3 que usa essa função. 

4. Qual a diferença entre uma subrotina e uma função ? 

 Funções - II 

Function fex5a() 
    fex5a = 10 
    Exit Function 
    fex5b = 20 
End Function 
  
  
  
Function fex5b() 
    Dim i As Integer 
    fex5b = 99 
    For i = 1 To 5 
        If i Mod 2 = 0 Then 
            Exit For 
        Else 
            fex5b = i ^ 2 
        End If 
    Next i 
End Function 
  
Function fex5c() 
    Dim i As Integer 
    fex5c = "X" 
    For i = 1 To 5 
        fex5c = fex5c & i 
    Next i 
    fex5c = "Eis eu." 
End Function 
  
Sub fex8() 
    Dim n 
    n = f_cint(Range("B2:B7"), 50, 99) 
End Sub 

Sub fex6() 
    Dim n As Integer 
    Dim r As Range 
     
    Set r = Range("B2:B7") 
    n = f_counta(r) 
    MsgBox "No grupo B2:C7 -> " & n 
  
    n = f_counta(Range("C2:C7")) 
    MsgBox "No grupo C2:C7 -> " & n 
  
    n = f_counta(Selection) 
    MsgBox "No grupo " & Selection.Address 
& " -> " & n 
  
    If f_counta(ActiveCell) = 1 Then 
        MsgBox "A celula activa não está 
vazia." 
    Else 
        MsgBox "A célula activa está 
vazia." 
    End If 
  
End Sub 
  
 
Function f_counta(r As Range) 
    Dim c As Range 
    Dim n As Integer 
     
    n = 0 
    For Each c In r 
        If IsEmpty(c.Value) Then n = n + 1 
    Next 
    f_counta = r.Count - n 
End Function 

1. Qual o resultado de cada um das funções fex5a, fex5b e fex5c ? 
Em geral, de que forma é estabelecido o resultado de uma função? 

2. A função f_counta( range) pretende ser parecida com a função COUNTA do Excel. 
Explique o efeito da subrotina fex6 que explora esta função. 

3. Faça uma função que receba dois argumentos - o primeiro um grupo de células (ou seja, um Range), o 
segundo um valor - e dê como resultado o número de vezes que esse valor aparece nas células do 
range. 

4. Faça a função f_cint de modo que a subrotina fex8 apresente como resultado o número de células do 
grupo B2:B7 com valores entre 50 e 99. 

 Funções. Utilização na folha de cálculo 

Function wks1(x) 
    MsgBox "Recebi o valor -> " & x 
    wks1 = x + 1 
End Function 
  
  
  
Function wks2(x, y) 
    wks2 = (x + y) / 2 
End Function 
 

1. Insira as fórmulas 
B2:=fex1() 
B3:=2*fex1()  . 
Confirme os resultados. 

2. Insira a fórmula 
B4:=wks1(33). 
Explique o efeito. 

3. Insira um número qualquer na célula C5 e a fórmula B5:=wks1(C5) . 
O que contece quando altera o valor de C5 ? 

4. Experimente no Excel a função wks2. Quais das seguintes chamadas são aceitáveis? 
a) =wks2( 3; 5) 
b) =wks2 (A1; A2) 
c) =wks2 ( A1+A2) 
d) =wks2 ( A1; "X") 
e) =wks2 (A1; "2") 
f) =wks2 ( A1; wks2(A1;A2)) 

Qual o resultado quando as chamadas não são "aceitáveis"? 

5. Verifique que as funções já feitas aparecem no fx. 

 Funções. Utilização na folha de cálculo 

Function wks11(x, tb As Range, ncol As Integer) 
    Dim i As Integer 
    Dim c As Range 
    For i = 1 To tb.Rows.Count 
        Set c = tb.Cells(i, 1) 
        If c.Value = x Then 
            wks11 = c.Offset(0, ncol - 1) 
            Exit Function 
        End If 
    Next 
    wks11 = "?" 
End Function 

1. Faça uma função F_SUM que se possa usar na seguinte fórmula 
=F_SUM(B2:C4) 
dando o mesmo resultado que a função SUM do Excel com o mesmo argumento. 

2. Faça uma função F_CONC que se possa usar na seguinte fórmula 
=F_CONC(B2:B5) 
dando como resultado a concatenação (ie o mesmo que o operador &) de todas as células do grupo. 

3. Faça uma função F_APROV que se possa usar na seguinte fórmula: 
=IF ( F_APROV(B1;C1) ; "Aprovado";"Reprovado") 
dando "Aprovado" se a média das notas B1, C1 der para passar e "Reprovado" caso contrário. 

4. Faça uma função que se possa usar na seguintes fórmula: 
=F_COUNTI(C2:C9;50;99) 
dando o número de valores do grupo compreendidos entre 50 e 99. 

5. Faça uma função F_AVIF que se possa usar na fórmula: 
=F_AVIF(C2:C9;"X";D2:D9) 
dando o mesmo resultado que 
=SUMIF(C2:C9;"X";D2:D9) / COUNTIF(C2:C9;"X") 

6. A função wks11 faz lembrar que função do Excel? 


